Chapter 2

Instructions: Language of the Computer
Instruction Set

- The repertoire of instructions of a computer
- Different computers have different instruction sets
  - But with many aspects in common
- Early computers had very simple instruction sets
  - Simplified implementation
- Many modern computers also have simple instruction sets
The MIPS Instruction Set

- Used as the example throughout the book
- Stanford MIPS commercialized by MIPS Technologies (www.mips.com)
- Large share of embedded core market
  - Applications in consumer electronics, network/storage equipment, cameras, printers, …
- Typical of many modern ISAs
  - See MIPS Reference Data tear-out card, and Appendixes B and E


## MIPS Operands

<table>
<thead>
<tr>
<th>Name</th>
<th>Example</th>
<th>Comments</th>
</tr>
</thead>
<tbody>
<tr>
<td>32 registers</td>
<td>$s0-$s7, $t0-$t9, $zero,</td>
<td>Fast locations for data. In MIPS, data must be in registers to perform arithmetic, register $zero always equals 0, and register $at is reserved by the assembler to handle large constants.</td>
</tr>
<tr>
<td></td>
<td>$a0-$a3, $v0-$v1, $gp, $fp,</td>
<td></td>
</tr>
<tr>
<td></td>
<td>$sp, $ra, $at</td>
<td></td>
</tr>
<tr>
<td>$2^{30}$ memory</td>
<td>Memory[0], Memory[4],...,</td>
<td>Accessed only by data transfer instructions. MIPS uses byte addresses, so sequential word addresses differ by 4. Memory holds data structures, arrays, and spilled registers.</td>
</tr>
<tr>
<td>words</td>
<td>Memory[4294967292]</td>
<td></td>
</tr>
</tbody>
</table>
### MIPS Instructions

<table>
<thead>
<tr>
<th>Category</th>
<th>Instruction</th>
<th>Example</th>
<th>Meaning</th>
<th>Comments</th>
</tr>
</thead>
<tbody>
<tr>
<td>Arithmetic</td>
<td>add</td>
<td>add $s1,$s2,$s3</td>
<td>$s1 = $s2 + $s3</td>
<td>Three register operands</td>
</tr>
<tr>
<td></td>
<td>subtract</td>
<td>sub $s1,$s2,$s3</td>
<td>$s1 = $s2 - $s3</td>
<td>Three register operands</td>
</tr>
<tr>
<td></td>
<td>add immediate</td>
<td>addi $s1,$s2,20</td>
<td>$s1 = $s2 + 20</td>
<td>Used to add constants</td>
</tr>
<tr>
<td>Data transfer</td>
<td>load word</td>
<td>lw $s1,20($s2)</td>
<td>$s1 = Memory[$s2 + 20]</td>
<td>Word from memory to register</td>
</tr>
<tr>
<td></td>
<td>store word</td>
<td>sw $s1,20($s2)</td>
<td>Memory[$s2 + 20] = $s1</td>
<td>Word from register to memory</td>
</tr>
<tr>
<td></td>
<td>load half</td>
<td>lh $s1,20($s2)</td>
<td>$s1 = Memory[$s2 + 20]</td>
<td>Halfword memory to register</td>
</tr>
<tr>
<td></td>
<td>load half unsigned</td>
<td>lhu $s1,20($s2)</td>
<td>$s1 = Memory[$s2 + 20]</td>
<td>Halfword memory to register</td>
</tr>
<tr>
<td></td>
<td>store half</td>
<td>sh $s1,20($s2)</td>
<td>Memory[$s2 + 20] = $s1</td>
<td>Halfword register to memory</td>
</tr>
<tr>
<td></td>
<td>load byte</td>
<td>lb $s1,20($s2)</td>
<td>$s1 = Memory[$s2 + 20]</td>
<td>Byte from memory to register</td>
</tr>
<tr>
<td></td>
<td>load byte unsigned</td>
<td>lbu $s1,20($s2)</td>
<td>$s1 = Memory[$s2 + 20]</td>
<td>Byte from memory to register</td>
</tr>
<tr>
<td></td>
<td>store byte</td>
<td>sb $s1,20($s2)</td>
<td>Memory[$s2 + 20] = $s1</td>
<td>Byte from register to memory</td>
</tr>
<tr>
<td></td>
<td>load linked word</td>
<td>ll $s1,20($s2)</td>
<td>$s1 = Memory[$s2 + 20]</td>
<td>Load word as 1st half of atomic swap</td>
</tr>
<tr>
<td></td>
<td>store condition. word</td>
<td>sc $s1,20($s2)</td>
<td>Memory[$s2+20]=$s1;$s1=0 or 1</td>
<td>Store word as 2nd half of atomic swap</td>
</tr>
<tr>
<td></td>
<td>load upper immedi.</td>
<td>lui $s1,20</td>
<td>$s1 = 20 * 2^16</td>
<td>Loads constant in upper 16 bits</td>
</tr>
</tbody>
</table>
### MIPS Instructions (cont’d)

<table>
<thead>
<tr>
<th>Logical</th>
<th>and</th>
<th>and $s1,$s2,$s3</th>
<th>$s1 = $s2 &amp; $s3</th>
<th>Three reg. operands; bit-by-bit AND</th>
</tr>
</thead>
<tbody>
<tr>
<td></td>
<td>or</td>
<td>or $s1,$s2,$s3</td>
<td>$s1 = $s2</td>
<td>$s3</td>
</tr>
<tr>
<td></td>
<td>nor</td>
<td>nor $s1,$s2,$s3</td>
<td>$s1 = ~($s2</td>
<td>$s3)</td>
</tr>
<tr>
<td></td>
<td>and immediate</td>
<td>andi $s1,$s2,20</td>
<td>$s1 = $s2 &amp; 20</td>
<td>Bit-by-bit AND reg with constant</td>
</tr>
<tr>
<td></td>
<td>or immediate</td>
<td>ori $s1,$s2,20</td>
<td>$s1 = $s2</td>
<td>20</td>
</tr>
<tr>
<td></td>
<td>shift left logical</td>
<td>sll $s1,$s2,10</td>
<td>$s1 = $s2 &lt;&lt; 10</td>
<td>Shift left by constant</td>
</tr>
<tr>
<td></td>
<td>shift right logical</td>
<td>srl $s1,$s2,10</td>
<td>$s1 = $s2 &gt;&gt; 10</td>
<td>Shift right by constant</td>
</tr>
<tr>
<td>Conditional branch</td>
<td>branch on equal</td>
<td>beq $s1,$s2,25</td>
<td>if($s1 == $s2) go to PC + 4 + 100</td>
<td>Equal test; PC-relative branch</td>
</tr>
<tr>
<td></td>
<td>branch on not equal</td>
<td>bne $s1,$s2,25</td>
<td>if($s1!= $s2) go to PC + 4 + 100</td>
<td>Not equal test; PC-relative</td>
</tr>
<tr>
<td></td>
<td>set on less than</td>
<td>slt $s1,$s2,$s3</td>
<td>if($s2 &lt; $s3) $s1 = 1; else $s1 = 0</td>
<td>Compare less than; for beq, bne</td>
</tr>
<tr>
<td></td>
<td>set on less than unsigned</td>
<td>sltu $s1,$s2,$s3</td>
<td>if($s2 &lt; $s3) $s1 = 1; else $s1 = 0</td>
<td>Compare less than unsigned</td>
</tr>
<tr>
<td></td>
<td>set less than immediate</td>
<td>slti $s1,$s2,20</td>
<td>if($s2 &lt; 20) $s1 = 1; else $s1 = 0</td>
<td>Compare less than constant</td>
</tr>
<tr>
<td></td>
<td>set less than immediate unsigned</td>
<td>sltiu $s1,$s2,20</td>
<td>if($s2 &lt; 20) $s1 = 1; else $s1 = 0</td>
<td>Compare less than constant unsigned</td>
</tr>
<tr>
<td>Unconditional jump</td>
<td>jump</td>
<td>j 2500</td>
<td>go to 10000</td>
<td>Jump to target address</td>
</tr>
<tr>
<td></td>
<td>jump register</td>
<td>jr $ra</td>
<td>go to $ra</td>
<td>For switch, procedure return</td>
</tr>
<tr>
<td></td>
<td>jump and link</td>
<td>jal 2500</td>
<td>$ra = PC + 4; go to 10000</td>
<td>For procedure call</td>
</tr>
</tbody>
</table>
Arithmetic Operations

- Add and subtract, three operands
  - Two sources and one destination
  
  \[
  \text{add } a, b, c \quad \# \ a \ \text{gets} \ b + c
  \]

- All arithmetic operations have this form

*Design Principle 1: Simplicity favours regularity*

  - Regularity makes implementation simpler
  - Simplicity enables higher performance at lower cost
Arithmetic Example

- C code:
  
  \[ f = (g + h) - (i + j); \]

- Compiled MIPS code:
  
  ```
  add t0, g, h   # temp t0 = g + h
  add t1, i, j   # temp t1 = i + j
  sub f, t0, t1  # f = t0 - t1
  ```
Register Operands

- Arithmetic instructions use register operands
- MIPS has a $32 \times 32$-bit register file
  - Use for frequently accessed data
  - Numbered 0 to 31
  - 32-bit data called a “word”
- Assembler names
  - $t0$, $t1$, …, $t9$ for temporary values
  - $s0$, $s1$, …, $s7$ for saved variables
- Design Principle 2: Smaller is faster
  - c.f. main memory: millions of locations
Register Operand Example

- C code:
  
  \[ f = (g + h) - (i + j); \]

  - \( f, \ldots, j \) in \( $s0, \ldots, $s4 \)

- Compiled MIPS code:

  - add \( $t0, $s1, $s2 \)
  - add \( $t1, $s3, $s4 \)
  - sub \( $s0, $t0, $t1 \)
Memory Operands

- Main memory used for composite data
  - Arrays, structures, dynamic data
- To apply arithmetic operations
  - Load values from memory into registers
  - Store result from register to memory
- Memory is byte addressed
  - Each address identifies an 8-bit byte
- Words are aligned in memory
  - Address must be a multiple of 4
- MIPS is Big Endian
  - Most-significant byte at least address of a word
  - c.f. Little Endian: least-significant byte at least address
Memory Operand Example 1

- **C code:**
  
  ```c
  g = h + A[8];
  ```
  
  - `g` in `$s1`, `h` in `$s2`, base address of `A` in `$s3`

- **Compiled MIPS code:**
  
  ```mips
  lw  $t0, 32($s3)    # load word
  add $s1, $s2, $t0
  ```
  
  - Index 8 requires offset of 32
  - 4 bytes per word
  
  ![Offset and Base Register Diagram]
Memory Operand Example 2

- **C code:**
  
  ```c
  ```
  
  - h in $s2, base address of A in $s3

- **Compiled MIPS code:**
  
  ```mips
  lw $t0, 32($s3)    # load word
  add $t0, $s2, $t0
  sw $t0, 48($s3)    # store word
  ```
Registers vs. Memory

- Registers are faster to access than memory
- Operating on memory data requires loads and stores
  - More instructions to be executed
- Compiler must use registers for variables as much as possible
  - Only spill to memory for less frequently used variables
  - Register optimization is important!
Immediate Operands

- Constant data specified in an instruction
  
  ```
  addi $s3, $s3, 4
  ```

- No subtract immediate instruction
  
  - Just use a negative constant
    
    ```
    addi $s2, $s1, -1
    ```

- *Design Principle 3*: Make the common case fast
  
  - Small constants are common
  
  - Immediate operand avoids a load instruction
The Constant Zero

- MIPS register 0 ($zero) is the constant 0
  - Cannot be overwritten
- Useful for common operations
  - E.g., move between registers
    add $t2, $s1, $zero
Unsigned Binary Integers

- Given an n-bit number
  \[ x = x_{n-1}2^{n-1} + x_{n-2}2^{n-2} + \cdots + x_12^1 + x_02^0 \]

- Range: 0 to \( +2^n - 1 \)

- Example
  - 0000 0000 0000 0000 0000 0000 0000 1011₂
    \[ = 0 + \cdots + 1 \times 2^3 + 0 \times 2^2 + 1 \times 2^1 + 1 \times 2^0 \]
    \[ = 0 + \cdots + 8 + 0 + 2 + 1 = 11_{10} \]

- Using 32 bits
  - 0 to \( +4,294,967,295 \)
2s-Complement Signed Integers

Given an n-bit number

\[ x = -x_{n-1}2^{n-1} + x_{n-2}2^{n-2} + \cdots + x_12^1 + x_02^0 \]

Range: \(-2^{n-1}\) to \(+2^{n-1} - 1\)

Example

\[ \begin{array}{c}
1111 \ 1111 \ 1111 \ 1111 \ 1111 \ 1111 \ 1111 \ 1100_2 \\
= -1 \times 2^{31} + 1 \times 2^{30} + \cdots + 1 \times 2^2 + 0 \times 2^1 + 0 \times 2^0 \\
= -2,147,483,648 + 2,147,483,644 = -4_{10}
\end{array} \]

Using 32 bits

\(-2,147,483,648\) to \(+2,147,483,647\)
2s-Complement Signed Integers

- Bit 31 is sign bit
  - 1 for negative numbers
  - 0 for non-negative numbers
- \(-(-2^{n-1})\) can’t be represented
- Non-negative numbers have the same unsigned and 2s-complement representation
- Some specific numbers
  - 0: 0000 0000 ... 0000
  - \(-1\): 1111 1111 ... 1111
  - Most-negative: 1000 0000 ... 0000
  - Most-positive: 0111 1111 ... 1111
Signed Negation

- Complement and add 1
  - Complement means 1 → 0, 0 → 1

\[ x + x = 1111\ldots111_2 = -1 \]
\[ x + 1 = -x \]

- Example: negate +2
  - +2 = 0000 0000 ... 0010_2
  - –2 = 1111 1111 ... 1101_2 + 1
    \[ = 1111 1111 ... 1110_2 \]
Sign Extension

- Representing a number using more bits
  - Preserve the numeric value

- In MIPS instruction set
  - `addi`: extend immediate value
  - `lb, lh`: extend loaded byte/halfword
  - `beq, bne`: extend the displacement

- Replicate the sign bit to the left
  - c.f. unsigned values: extend with 0s

- Examples: 8-bit to 16-bit
  - +2: 0000 0010 => 0000 0000 0000 0010
  - –2: 1111 1110 => 1111 1111 1111 1110
Instructions are encoded in binary
- Called machine code

MIPS instructions
- Encoded as 32-bit instruction words
- Small number of formats encoding operation code (opcode), register numbers, …
- Regularity!

Register numbers
- $t0 – t7$ are reg’s 8 – 15
- $t8 – t9$ are reg’s 24 – 25
- $s0 – s7$ are reg’s 16 – 23
# MIPS R-format Instructions

<table>
<thead>
<tr>
<th>op</th>
<th>rs</th>
<th>rt</th>
<th>rd</th>
<th>shamt</th>
<th>funct</th>
</tr>
</thead>
<tbody>
<tr>
<td>6 bits</td>
<td>5 bits</td>
<td>5 bits</td>
<td>5 bits</td>
<td>5 bits</td>
<td>6 bits</td>
</tr>
</tbody>
</table>

## Instruction fields
- **op**: operation code (opcode)
- **rs**: first source register number
- **rt**: second source register number
- **rd**: destination register number
- **shamt**: shift amount (000000 for now)
- **funct**: function code (extends opcode)
### R-format Example

<table>
<thead>
<tr>
<th>op</th>
<th>rs</th>
<th>rt</th>
<th>rd</th>
<th>shamt</th>
<th>funct</th>
</tr>
</thead>
<tbody>
<tr>
<td>6 bits</td>
<td>5 bits</td>
<td>5 bits</td>
<td>5 bits</td>
<td>5 bits</td>
<td>6 bits</td>
</tr>
</tbody>
</table>

#### add \$t0, \$s1, \$s2

<table>
<thead>
<tr>
<th>special</th>
<th>$s1</th>
<th>$s2</th>
<th>$t0</th>
<th>0</th>
<th>add</th>
</tr>
</thead>
<tbody>
<tr>
<td>0</td>
<td>17</td>
<td>18</td>
<td>8</td>
<td>0</td>
<td>32</td>
</tr>
</tbody>
</table>

| 000000 | 10001 | 10010 | 01000 | 00000 | 100000 |

\[
00000010001100100100000000100000_2 = 02324020_{16}
\]
Hexadecimal

- Base 16
  - Compact representation of bit strings
  - 4 bits per hex digit

<p>| | | | | | | | |</p>
<table>
<thead>
<tr>
<th></th>
<th></th>
<th></th>
<th></th>
<th></th>
<th></th>
<th></th>
<th></th>
</tr>
</thead>
<tbody>
<tr>
<td>0</td>
<td>0000</td>
<td>4</td>
<td>0100</td>
<td>8</td>
<td>1000</td>
<td>c</td>
<td>1100</td>
</tr>
<tr>
<td>1</td>
<td>0001</td>
<td>5</td>
<td>0101</td>
<td>9</td>
<td>1001</td>
<td>d</td>
<td>1101</td>
</tr>
<tr>
<td>2</td>
<td>0010</td>
<td>6</td>
<td>0110</td>
<td>a</td>
<td>1010</td>
<td>e</td>
<td>1110</td>
</tr>
<tr>
<td>3</td>
<td>0011</td>
<td>7</td>
<td>0111</td>
<td>b</td>
<td>1011</td>
<td>f</td>
<td>1111</td>
</tr>
</tbody>
</table>

- Example: eca8 6420
  - 1110 1100 1010 1000 0110 0100 0010 0000
**MIPS I-format Instructions**

<table>
<thead>
<tr>
<th>op</th>
<th>rs</th>
<th>rt</th>
<th>constant or address</th>
</tr>
</thead>
<tbody>
<tr>
<td>6 bits</td>
<td>5 bits</td>
<td>5 bits</td>
<td>16 bits</td>
</tr>
</tbody>
</table>

- Immediate arithmetic and load/store instructions
  - rt: destination or source register number
  - Constant: $-2^{15}$ to $+2^{15} - 1$
  - Address: offset added to base address in rs

**Design Principle 4:** Good design demands good compromises
- Different formats complicate decoding, but allow 32-bit instructions uniformly
- Keep formats as similar as possible
Stored Program Computers

The BIG Picture

- Instructions represented in binary, just like data
- Instructions and data stored in memory
- Programs can operate on programs
  - e.g., compilers, linkers, …
- Binary compatibility allows compiled programs to work on different computers
  - Standardized ISAs
## Logical Operations

**Instructions for bitwise manipulation**

<table>
<thead>
<tr>
<th>Operation</th>
<th>C</th>
<th>Java</th>
<th>MIPS</th>
</tr>
</thead>
<tbody>
<tr>
<td>Shift left</td>
<td>&lt;&lt;</td>
<td>&lt;&lt;</td>
<td>sll</td>
</tr>
<tr>
<td>Shift right</td>
<td>&gt;&gt;</td>
<td>&gt;&gt;&gt;</td>
<td>srl</td>
</tr>
<tr>
<td>Bitwise AND</td>
<td>&amp;</td>
<td>&amp;</td>
<td>and, andi</td>
</tr>
<tr>
<td>Bitwise OR</td>
<td></td>
<td></td>
<td>or, ori</td>
</tr>
<tr>
<td>Bitwise NOT</td>
<td>~</td>
<td>~</td>
<td>nor</td>
</tr>
</tbody>
</table>

- Useful for extracting and inserting groups of bits in a word
Shift Operations

<table>
<thead>
<tr>
<th>op</th>
<th>rs</th>
<th>rt</th>
<th>rd</th>
<th>shamt</th>
<th>funct</th>
</tr>
</thead>
<tbody>
<tr>
<td>6 bits</td>
<td>5 bits</td>
<td>5 bits</td>
<td>5 bits</td>
<td>5 bits</td>
<td>6 bits</td>
</tr>
</tbody>
</table>

- **shamt**: how many positions to shift
- **Shift left logical**
  - Shift left and fill with 0 bits
  - `sll` by `i` bits multiplies by $2^i$
- **Shift right logical**
  - Shift right and fill with 0 bits
  - `srl` by `i` bits divides by $2^i$ (unsigned only)
AND Operations

- Useful to mask bits in a word
  - Select some bits, clear others to 0

and $t0, $t1, $t2

```
$t0  0000 0000 0000 0000 0000 0000 1100 0000
$t1  0000 0000 0000 0000 0011 1100 0000 0000
$t2  0000 0000 0000 0000 0000 1101 1100 0000
```
OR Operations

- Useful to include bits in a word
  - Set some bits to 1, leave others unchanged

or \( t0, t1, t2 \)

<p>| | | | | | | | | |</p>
<table>
<thead>
<tr>
<th></th>
<th></th>
<th></th>
<th></th>
<th></th>
<th></th>
<th></th>
<th></th>
<th></th>
</tr>
</thead>
<tbody>
<tr>
<td>( t2 )</td>
<td>0000</td>
<td>0000</td>
<td>0000</td>
<td>0000</td>
<td>0000</td>
<td>0000</td>
<td>1101</td>
<td>1100</td>
</tr>
<tr>
<td>( t1 )</td>
<td>0000</td>
<td>0000</td>
<td>0000</td>
<td>0000</td>
<td>0011</td>
<td>1100</td>
<td>0000</td>
<td>0000</td>
</tr>
<tr>
<td>( t0 )</td>
<td>0000</td>
<td>0000</td>
<td>0000</td>
<td>0000</td>
<td>0011</td>
<td>1101</td>
<td>1100</td>
<td>0000</td>
</tr>
</tbody>
</table>
NOT Operations

- Useful to invert bits in a word
  - Change 0 to 1, and 1 to 0
- MIPS has NOR 3-operand instruction
  - \( a \text{ NOR } b = \text{ NOT( } a \text{ OR } b \text{ )} \)

\[
nor \; \texttt{\$t0, \$t1, \$zero}
\]

Register 0: always read as zero

<table>
<thead>
<tr>
<th>$t1$</th>
<th>0000 0000 0000 0000 0011 1100 0000 0000</th>
</tr>
</thead>
<tbody>
<tr>
<td>$t0$</td>
<td>1111 1111 1111 1111 1100 0011 1111 1111</td>
</tr>
</tbody>
</table>
Conditional Operations

- Branch to a labeled instruction if a condition is true
  - Otherwise, continue sequentially
- beq rs, rt, L1
  - if (rs == rt) branch to instruction labeled L1;
- bne rs, rt, L1
  - if (rs != rt) branch to instruction labeled L1;
- j L1
  - unconditional jump to instruction labeled L1
Compiling If Statements

- **C code:**
  
  ```c
  if (i==j) f = g+h;
  else f = g-h;
  ```

  - `f, g, ...` in `$s0, $s1, ...`

- **Compiled MIPS code:**
  
  ```asm
  bne $s3, $s4, Else
  add $s0, $s1, $s2
  j Exit
  Else: sub $s0, $s1, $s2
  Exit: ...
  ```

  *Assembler calculates addresses*
Compiling Loop Statements

- C code:
  ```
  while (save[i] == k) i += 1;
  ```
  - i in $s3, k in $s5, address of save in $s6

- Compiled MIPS code:
  ```
  Loop: sll   $t1, $s3, 2
        add   $t1, $t1, $s6
        lw    $t0, 0($t1)
        bne   $t0, $s5, Exit
        addi  $s3, $s3, 1
        j     Loop
  Exit: ...
  ```
Basic Blocks

- A basic block is a sequence of instructions with
  - No embedded branches (except at end)
  - No branch targets (except at beginning)

- A compiler identifies basic blocks for optimization
- An advanced processor can accelerate execution of basic blocks
More Conditional Operations

- Set result to 1 if a condition is true
  - Otherwise, set to 0
- `slt rd, rs, rt`
  - if \((rs < rt)\) \(rd = 1\); else \(rd = 0\);
- `slti rt, rs, constant`
  - if \((rs < constant)\) \(rt = 1\); else \(rt = 0\);
- Use in combination with `beq`, `bne`
  
  `slt \$t0, \$s1, \$s2`  # if \(($s1 < \$s2$)\)
  
  `bne \$t0, \$zero, L`  # branch to \(L\)`
Branch Instruction Design

- Why not blt, bge, etc?
- Hardware for <, ≥, … slower than =, ≠
  - Combining with branch involves more work per instruction, requiring a slower clock
  - All instructions penalized!
- beq and bne are the common case
- This is a good design compromise
Signed vs. Unsigned

- Signed comparison: `slt`, `slti`
- Unsigned comparison: `sltu`, `sltui`

Example

- $s0 = 1111\ 1111\ 1111\ 1111\ 1111\ 1111\ 1111\ 1111$
- $s1 = 0000\ 0000\ 0000\ 0000\ 0000\ 0000\ 0000\ 0001$
- `slt $t0, $s0, $s1` # signed
  - $-1 < +1 \Rightarrow t0 = 1$
- `sltu $t0, $s0, $s1` # unsigned
  - $+4,294,967,295 > +1 \Rightarrow t0 = 0$
Procedure Calling

Steps required
1. Place parameters in registers
2. Transfer control to procedure
3. Acquire storage for procedure
4. Perform procedure’s operations
5. Place result in register for caller
6. Return to place of call
Register Usage

- $a0 – $a3: arguments (reg’s 4 – 7)
- $v0, $v1: result values (reg’s 2 and 3)
- $t0 – $t9: temporaries
  - Can be overwritten by callee
- $s0 – $s7: saved
  - Must be saved/restored by callee
- $gp: global pointer for static data (reg 28)
- $sp: stack pointer (reg 29)
- $fp: frame pointer (reg 30)
- $ra: return address (reg 31)
Procedure Call Instructions

- Procedure call: jump and link
  ```assembly
  jal ProcedureLabel
  ```
  - Address of following instruction put in $ra
  - Jumps to target address

- Procedure return: jump register
  ```assembly
  jr $ra
  ```
  - Copies $ra to program counter
  - Can also be used for computed jumps
    - e.g., for case/switch statements
Leaf Procedure Example

- C code:

  ```c
  int leaf_example (int g, h, i, j)
  {
    int f;
    f = (g + h) - (i + j);
    return f;
  }
  ```

- Arguments $g, ..., j$ in $a0, ..., a3$
- $f$ in $s0$ (hence, need to save $s0$ on stack)
- Result in $v0$
## Leaf Procedure Example

**MIPS code:**

<table>
<thead>
<tr>
<th>leaf_example:</th>
<th></th>
</tr>
</thead>
<tbody>
<tr>
<td><code>addi $sp, $sp, -4</code></td>
<td></td>
</tr>
<tr>
<td><code>sw $s0, 0($sp)</code></td>
<td></td>
</tr>
<tr>
<td><code>add $t0, $a0, $a1</code></td>
<td></td>
</tr>
<tr>
<td><code>add $t1, $a2, $a3</code></td>
<td></td>
</tr>
<tr>
<td><code>sub $s0, $t0, $t1</code></td>
<td></td>
</tr>
<tr>
<td><code>add $v0, $s0, $zero</code></td>
<td></td>
</tr>
<tr>
<td><code>lw $s0, 0($sp)</code></td>
<td></td>
</tr>
<tr>
<td><code>addi $sp, $sp, 4</code></td>
<td></td>
</tr>
<tr>
<td><code>jr $ra</code></td>
<td></td>
</tr>
</tbody>
</table>

- **Save $s0 on stack**
- **Procedure body**
- **Result**
- **Restore $s0**
- **Return**
Nested (non-leaf) Procedures

- Procedures that call other procedures
- For nested call, caller needs to save on the stack:
  - Its return address
  - Any arguments and temporaries needed after the call
- Restore from the stack after the call
Non-Leaf Procedure Example

- C code:
  ```c
  int fact (int n)
  {
    if (n < 1) return 1;
    else return n * fact(n - 1);
  }
  ```

- Argument n in $a0
- Result in $v0
Non-Leaf Procedure Example

MIPS code:

```
fact:
  addi $sp, $sp, -8       # adjust stack for 2 items
  sw   $ra, 4($sp)        # save return address
  sw   $a0, 0($sp)        # save argument
  slti $t0, $a0, 1        # test for n < 1
  beq  $t0, $zero, L1

  addi $v0, $zero, 1      # if so, result is 1
  addi $sp, $sp, 8        # pop 2 items from stack
  jr   $ra                # and return

L1:  addi $a0, $a0, -1    # else decrement n
    jal  fact             # recursive call

  lw   $a0, 0($sp)        # restore original n
  lw   $ra, 4($sp)        # and return address
  addi $sp, $sp, 8        # pop 2 items from stack
  mul  $v0, $a0, $v0      # multiply to get result
  jr   $ra                # and return
```
Local Data on the Stack

- Local data allocated by callee
  - e.g., C automatic variables
- Procedure frame (activation record)
  - Used by some compilers to manage stack storage

Diagram:
- High address
  - $fp$ →
  - $sp$ →
- Low address
  - a. $fp$ → $sp$ →
  - b. $fp$ →
    - Saved argument registers (if any)
    - Saved return address
    - Saved saved registers (if any)
    - Local arrays and structures (if any)
  - c. $fp$ → $sp$ →
Memory Layout

- **Text**: program code
- **Static data**: global variables
  - e.g., static variables in C, constant arrays and strings
  - $gp$ initialized to address allowing $\pm$ offsets into this segment
- **Dynamic data**: heap
  - E.g., malloc in C, new in Java
- **Stack**: automatic storage
Character Data

- Byte-encoded character sets
  - ASCII: 128 characters
    - 95 graphic, 33 control
  - Latin-1: 256 characters
    - ASCII, +96 more graphic characters
- Unicode: 32-bit character set
  - Used in Java, C++ wide characters, …
  - Most of the world’s alphabets, plus symbols
  - UTF-8, UTF-16: variable-length encodings
Byte/Halfword Operations

- Could use bitwise operations
- MIPS byte/halfword load/store
  - String processing is a common case
  - `lb rt, offset(rs)   lh rt, offset(rs)`
  - Sign extend to 32 bits in rt
  - `lbu rt, offset(rs)   lhu rt, offset(rs)`
  - Zero extend to 32 bits in rt
  - `sb rt, offset(rs)   sh rt, offset(rs)`
  - Store just rightmost byte/halfword
String Copy Example

- C code (naïve):
  - Null-terminated string
  ```c
  void strcpy (char x[], char y[])
  {
    int i;
    i = 0;
    while ((x[i]=y[i])!='\0')
      i += 1;
  }
  ```
  - Addresses of x, y in $a0, $a1
  - i in $s0
### String Copy Example

#### MIPS code:

<table>
<thead>
<tr>
<th>Instruction</th>
<th>Description</th>
</tr>
</thead>
<tbody>
<tr>
<td>addi $sp, $sp, -4</td>
<td># adjust stack for 1 item</td>
</tr>
<tr>
<td>sw $s0, 0($sp)</td>
<td># save $s0</td>
</tr>
<tr>
<td>add $s0, $zero, $zero</td>
<td># i = 0</td>
</tr>
<tr>
<td>L1: add $t1, $s0, $a1</td>
<td># addr of y[i] in $t1</td>
</tr>
<tr>
<td>lbu $t2, 0($t1)</td>
<td># $t2 = y[i]</td>
</tr>
<tr>
<td>add $t3, $s0, $a0</td>
<td># addr of x[i] in $t3</td>
</tr>
<tr>
<td>sb $t2, 0($t3)</td>
<td># x[i] = y[i]</td>
</tr>
<tr>
<td>beq $t2, $zero, L2</td>
<td># exit loop if y[i] == 0</td>
</tr>
<tr>
<td>addi $s0, $s0, 1</td>
<td># i = i + 1</td>
</tr>
<tr>
<td>j L1</td>
<td># next iteration of loop</td>
</tr>
<tr>
<td>L2: lw $s0, 0($sp)</td>
<td># restore saved $s0</td>
</tr>
<tr>
<td>addi $sp, $sp, 4</td>
<td># pop 1 item from stack</td>
</tr>
<tr>
<td>jr $ra</td>
<td># and return</td>
</tr>
</tbody>
</table>
32-bit Constants

- Most constants are small
  - 16-bit immediate is sufficient
- For the occasional 32-bit constant
  - `lui rt, constant`
    - Copies 16-bit constant to left 16 bits of rt
    - Clears right 16 bits of rt to 0

```
lui $s0, 61
ori $s0, $s0, 2304
```

<table>
<thead>
<tr>
<th>0000 0000 0111 1101</th>
<th>0000 0000 0000 0000</th>
</tr>
</thead>
<tbody>
<tr>
<td>0000 0000 0111 1101</td>
<td>0000 1001 0000 0000</td>
</tr>
</tbody>
</table>
Branch Addressing

- Branch instructions specify
  - Opcode, two registers, target address
- Most branch targets are near branch
  - Forward or backward

<table>
<thead>
<tr>
<th>op</th>
<th>rs</th>
<th>rt</th>
<th>constant or address</th>
</tr>
</thead>
<tbody>
<tr>
<td>6 bits</td>
<td>5 bits</td>
<td>5 bits</td>
<td>16 bits</td>
</tr>
</tbody>
</table>

- PC-relative addressing
  - Target address = PC + offset × 4
  - PC already incremented by 4 by this time
Jump Addressing

- Jump (j and jal) targets could be anywhere in text segment
  - Encode full address in instruction

(Pseudo)Direct jump addressing
- Target address = \( PC_{31...28} \times (\text{address} \times 4) \)
Target Addressing Example

- Loop code from earlier example
- Assume Loop at location 80000

<table>
<thead>
<tr>
<th>Instruction</th>
<th>Address</th>
<th>Target</th>
</tr>
</thead>
<tbody>
<tr>
<td>sll $t1, $s3, 2</td>
<td>80000</td>
<td>0 0 19 9 2 0</td>
</tr>
<tr>
<td>add $t1, $t1, $s6</td>
<td>80004</td>
<td>0 9 22 9 0 32</td>
</tr>
<tr>
<td>lw $t0, 0($t1)</td>
<td>80008</td>
<td>35 9 8 0</td>
</tr>
<tr>
<td>bne $t0, $s5, Exit</td>
<td>80012</td>
<td>5 8 21 2</td>
</tr>
<tr>
<td>addi $s3, $s3, 1</td>
<td>80016</td>
<td>8 19 19 1</td>
</tr>
<tr>
<td>j Loop</td>
<td>80020</td>
<td>2 20000</td>
</tr>
</tbody>
</table>

Exit: ...

80024
Branching Far Away

- If branch target is too far to encode with 16-bit offset, assembler rewrites the code.

Example

```
beq $s0,$s1, L1
↓
bne $s0,$s1, L2
    j L1
L2: ...
```
Addressing Mode Summary

1. Immediate addressing
   \[ \text{op rs rt Immediate} \]

2. Register addressing
   \[ \text{op rs rt rd . . . funct} \]

3. Base addressing
   \[ \text{op rs rt Address} \]

4. PC-relative addressing
   \[ \text{op rs rt Address} \]

5. Pseudodirect addressing
   \[ \text{op Address} \]

Chapter 2 — Instructions: Language of the Computer — 59
Decoding Machine Language

What is the assembly language statement corresponding to this machine instruction?

00af8020hex

add $s0, $a1, $t7
## Instruction Formats and Register Names

### R-format
- **Fields**: 
  - `op`: 6 bits
  - `rs`: 5 bits
  - `rt`: 5 bits
  - `rd`: 5 bits
  - `shamt`: 5 bits
  - `funct`: 6 bits
- **Comments**: Arithmetic instruction format

### I-format
- **Fields**: 
  - `op`: 6 bits
  - `rs`: 5 bits
  - `rt`: 5 bits
  - `address/Immediate`: 6 bits
- **Comments**: Transfer, branch, imm. format

### J-format
- **Fields**: 
  - `op`: 6 bits
  - `target address`: 6 bits
- **Comments**: Jump instruction format

### Register Table
<table>
<thead>
<tr>
<th>Name</th>
<th>Register number</th>
<th>Usage</th>
<th>Preserved on call?</th>
</tr>
</thead>
<tbody>
<tr>
<td>$zero</td>
<td>0</td>
<td>The constant value 0</td>
<td>n.a.</td>
</tr>
<tr>
<td>$v0–$v1</td>
<td>2–3</td>
<td>Values for results and expression evaluation</td>
<td>no</td>
</tr>
<tr>
<td>$a0–$a3</td>
<td>4–7</td>
<td>Arguments</td>
<td>no</td>
</tr>
<tr>
<td>$t0–$t7</td>
<td>8–15</td>
<td>Temporaries</td>
<td>no</td>
</tr>
<tr>
<td>$s0–$s7</td>
<td>16–23</td>
<td>Saved</td>
<td>yes</td>
</tr>
<tr>
<td>$t8–$t9</td>
<td>24–25</td>
<td>More temporaries</td>
<td>no</td>
</tr>
<tr>
<td>$gp</td>
<td>28</td>
<td>Global pointer</td>
<td>yes</td>
</tr>
<tr>
<td>$sp</td>
<td>29</td>
<td>Stack pointer</td>
<td>yes</td>
</tr>
<tr>
<td>$fp</td>
<td>30</td>
<td>Frame pointer</td>
<td>yes</td>
</tr>
<tr>
<td>$ra</td>
<td>31</td>
<td>Return address</td>
<td>yes</td>
</tr>
</tbody>
</table>
## Instruction Encoding

<table>
<thead>
<tr>
<th>op(31:26)</th>
<th>28–26</th>
<th>31–29</th>
</tr>
</thead>
<tbody>
<tr>
<td></td>
<td>0(000)</td>
<td>1(001)</td>
</tr>
<tr>
<td>0(000)</td>
<td>R-format</td>
<td>Bltz/gez</td>
</tr>
<tr>
<td>1(001)</td>
<td>add immediate</td>
<td>addiu</td>
</tr>
<tr>
<td>2(010)</td>
<td>TLB</td>
<td>FlPt</td>
</tr>
<tr>
<td>3(011)</td>
<td></td>
<td></td>
</tr>
<tr>
<td>4(100)</td>
<td>load byte</td>
<td>load half</td>
</tr>
<tr>
<td>5(101)</td>
<td>store byte</td>
<td>store half</td>
</tr>
<tr>
<td>6(110)</td>
<td>load linked word</td>
<td>lw $c1$</td>
</tr>
<tr>
<td>7(111)</td>
<td>store cond. word</td>
<td>sw $c1$</td>
</tr>
</tbody>
</table>
### Instruction Encoding (cont’d)

<table>
<thead>
<tr>
<th>op(31:26)=000000 (R-format), funct(5:0)</th>
<th>0(000)</th>
<th>1(001)</th>
<th>2(010)</th>
<th>3(011)</th>
<th>4(100)</th>
<th>5(101)</th>
<th>6(110)</th>
<th>7(111)</th>
</tr>
</thead>
<tbody>
<tr>
<td>2-0</td>
<td>0(000)</td>
<td>1(001)</td>
<td>2(010)</td>
<td>3(011)</td>
<td>4(100)</td>
<td>5(101)</td>
<td>6(110)</td>
<td>7(111)</td>
</tr>
<tr>
<td>5-3</td>
<td>shift left logical</td>
<td>shift right logical</td>
<td>sra</td>
<td>sllv</td>
<td>srlv</td>
<td>sra v</td>
<td></td>
<td></td>
</tr>
<tr>
<td>0(000)</td>
<td>jump register</td>
<td>jalr</td>
<td></td>
<td></td>
<td>syscall</td>
<td>break</td>
<td></td>
<td></td>
</tr>
<tr>
<td>1(001)</td>
<td>mfhi</td>
<td>mthi</td>
<td>mflo</td>
<td>mtlo</td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>2(010)</td>
<td>mult</td>
<td>multu</td>
<td>div</td>
<td>divu</td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>3(011)</td>
<td>add</td>
<td>addu</td>
<td>subtract</td>
<td>subu</td>
<td>and</td>
<td>or</td>
<td>xor</td>
<td>not or (nor)</td>
</tr>
<tr>
<td>4(100)</td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>5(101)</td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>6(110)</td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>7(111)</td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
</tbody>
</table>
Synchronization in MIPS

- Atomically read and modify a memory location
- Load linked: `ll rt, offset(rs)`
- Store conditional: `sc rt, offset(rs)`
  - Succeeds if location not changed since the `ll`
    - Returns 1 in rt
  - Fails if location is changed
    - Returns 0 in rt
- Example: atomic swap (to test/set lock variable)
  again: `addi $t0,$zero, 1` ; copy exchange value
  `ll $t1,0($s1)` ; load linked
  `sc $t0,0($s1)` ; store conditional
  `beq $t0,$zero,again` ; branch store fails
  `add $s4,$zero,$t1` ; put load value in $s4
Many compilers produce object modules directly.

C program

Compiler

Assembly language program

Assembler

Object: Machine language module

Object: Library routine (machine language)

Linker

Executable: Machine language program

Loader

Memory

Static linking

Chapter 2 — Instructions: Language of the Computer — 65
Assembler Pseudoinstructions

- Most assembler instructions represent machine instructions one-to-one
- Pseudoinstructions: figments of the assembler’s imagination

- move $t0, $t1  →  add $t0, $zero, $t1
- blt $t0, $t1, L  →  slt $at, $t0, $t1
- bne $at, $zero, L

- $at (register 1): assembler temporary
Producing an Object Module

- Assembler (or compiler) translates program into machine instructions
- Provides information for building a complete program from the pieces
  - Header: described contents of object module
  - Text segment: translated instructions
  - Static data segment: data allocated for the life of the program
  - Relocation info: for contents that depend on absolute location of loaded program
  - Symbol table: global definitions and external refs
  - Debug info: for associating with source code
Linking Object Modules

- Produces an executable image
  1. Merges segments
  2. Resolve labels (determine their addresses)
  3. Patch location-dependent and external refs

- Could leave location dependencies for fixing by a relocating loader
  - But with virtual memory, no need to do this
  - Program can be loaded into absolute location in virtual memory space
Loading a Program

- Load from image file on disk into memory
  1. Read header to determine segment sizes
  2. Create virtual address space
  3. Copy text and initialized data into memory
     - Or set page table entries so they can be faulted in
  4. Set up arguments on stack
  5. Initialize registers (including $sp, $fp, $gp)
  6. Jump to startup routine
     - Copies arguments to $a0, … and calls main
     - When main returns, do exit syscall
Dynamic Linking

- Only link/load library procedure when it is called
  - Requires procedure code to be relocatable
  - Avoids image bloat caused by static linking of all (transitively) referenced libraries
  - Automatically picks up new library versions
Starting Java Applications

- Java program
- Class files (Java bytecodes)
- Java Virtual Machine
- Compiled Java methods (machine language)
- Interprets bytecodes
- Simple portable instruction set for the JVM
- Compiler
- Java library routines (machine language)

Compiles bytecodes of “hot” methods into native code for host machine.
C Sort Example

- Illustrates use of assembly instructions for a C bubble sort function
- Swap procedure (leaf)
  ```c
  void swap(int v[], int k)
  {
    int temp;
    temp = v[k];
    v[k] = v[k+1];
    v[k+1] = temp;
  }
  ```
- v in $a0, k in $a1, temp in $t0
The Procedure Swap

<table>
<thead>
<tr>
<th>Instruction</th>
<th>Description</th>
</tr>
</thead>
<tbody>
<tr>
<td><code>sll $t1, $a1, 2</code></td>
<td>$t1 = k \times 4</td>
</tr>
<tr>
<td><code>add $t1, $a0, $t1</code></td>
<td>$t1 = v + (k \times 4)$</td>
</tr>
<tr>
<td></td>
<td>(# (address of v[k]))</td>
</tr>
<tr>
<td><code>lw $t0, 0($t1)</code></td>
<td>$t0$ (temp) = v[k]</td>
</tr>
<tr>
<td><code>lw $t2, 4($t1)</code></td>
<td>$t2 = v[k+1]$</td>
</tr>
<tr>
<td><code>sw $t2, 0($t1)</code></td>
<td>v[k] = $t2 (v[k+1])</td>
</tr>
<tr>
<td><code>sw $t0, 4($t1)</code></td>
<td>v[k+1] = $t0 (temp)</td>
</tr>
<tr>
<td><code>jr $ra</code></td>
<td># return to calling routine</td>
</tr>
</tbody>
</table>
The Sort Procedure in C

- Non-leaf (calls swap)
  ```c
  void sort (int v[], int n)
  {
    int i, j;
    for (i = 0; i < n; i += 1) {
      for (j = i - 1;
        j >= 0 && v[j] > v[j + 1];
        j -= 1) {
        swap(v,j);
      }
    }
  }
  ```

- v in $a0, n in $a1, i in $s0, j in $s1
The Procedure Body

<table>
<thead>
<tr>
<th>Move params</th>
<th>Outer loop</th>
<th>Inner loop</th>
<th>Pass params &amp; call</th>
<th>Inner loop</th>
<th>Outer loop</th>
</tr>
</thead>
<tbody>
<tr>
<td>move $s2, $a0</td>
<td># save $a0 into $s2</td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>move $s3, $a1</td>
<td># save $a1 into $s3</td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>move $s0, $zero</td>
<td># i = 0</td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>for1tst: slt $t0, $s0, $s3</td>
<td># $t0 = 0 if $s0 ≥ $s3 (i ≥ n)</td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>beq $t0, $zero, exit1</td>
<td># go to exit1 if $s0 ≥ $s3 (i ≥ n)</td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>addi $s1, $s0, -1</td>
<td># j = i - 1</td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>for2tst: slti $t0, $s1, 0</td>
<td># $t0 = 1 if $s1 &lt; 0 (j &lt; 0)</td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>bne $t0, $zero, exit2</td>
<td># go to exit2 if $s1 &lt; 0 (j &lt; 0)</td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>sll $t1, $s1, 2</td>
<td># $t1 = j * 4</td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>add $t2, $s2, $t1</td>
<td># $t2 = v + (j * 4)</td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>lw $t3, 0($t2)</td>
<td># $t3 = v[j]</td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>lw $t4, 4($t2)</td>
<td># $t4 = v[j + 1]</td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>slt $t0, $t4, $t3</td>
<td># $t0 = 0 if $t4 ≥ $t3</td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>beq $t0, $zero, exit2</td>
<td># go to exit2 if $t4 ≥ $t3</td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>move $a0, $s2</td>
<td># 1st param of swap is v (old $a0)</td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>move $a1, $s1</td>
<td># 2nd param of swap is j</td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>jal swap</td>
<td># call swap procedure</td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>addi $s1, $s1, -1</td>
<td># j = j - 1</td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>j for2tst</td>
<td># jump to test of inner loop</td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>exit2: addi $s0, $s0, 1</td>
<td># i = i + 1</td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
<tr>
<td>j for1tst</td>
<td># jump to test of outer loop</td>
<td></td>
<td></td>
<td></td>
<td></td>
</tr>
</tbody>
</table>
### The Full Procedure

<table>
<thead>
<tr>
<th>Instruction</th>
<th>Description</th>
</tr>
</thead>
<tbody>
<tr>
<td><code>sort:</code></td>
<td><code>addi $sp,$sp, -20</code> # make room on stack for 5 registers</td>
</tr>
<tr>
<td><code>sw $ra, 16($sp)</code></td>
<td># save $ra on stack</td>
</tr>
<tr>
<td><code>sw $s3,12($sp)</code></td>
<td># save $s3 on stack</td>
</tr>
<tr>
<td><code>sw $s2, 8($sp)</code></td>
<td># save $s2 on stack</td>
</tr>
<tr>
<td><code>sw $s1, 4($sp)</code></td>
<td># save $s1 on stack</td>
</tr>
<tr>
<td><code>sw $s0, 0($sp)</code></td>
<td># save $s0 on stack</td>
</tr>
<tr>
<td></td>
<td># procedure body</td>
</tr>
<tr>
<td><code>exit1:</code></td>
<td><code>lw $s0, 0($sp)</code> # restore $s0 from stack</td>
</tr>
<tr>
<td><code>lw $s1, 4($sp)</code></td>
<td># restore $s1 from stack</td>
</tr>
<tr>
<td><code>lw $s2, 8($sp)</code></td>
<td># restore $s2 from stack</td>
</tr>
<tr>
<td><code>lw $s3,12($sp)</code></td>
<td># restore $s3 from stack</td>
</tr>
<tr>
<td><code>lw $ra,16($sp)</code></td>
<td># restore $ra from stack</td>
</tr>
<tr>
<td><code>addi $sp,$sp, 20</code></td>
<td># restore stack pointer</td>
</tr>
<tr>
<td><code>jr $ra</code></td>
<td># return to calling routine</td>
</tr>
</tbody>
</table>
Effect of Compiler Optimization

Compiled with gcc for Pentium 4 under Linux

- Relative Performance
- Instruction count
- Clock Cycles
- CPI

Chapter 2 — Instructions: Language of the Computer — 78
Effect of Language and Algorithm

![Bubblesort Relative Performance](image1)

![Quicksort Relative Performance](image2)

![Quicksort vs. Bubblesort Speedup](image3)
Lessons Learnt

- Instruction count and CPI are not good performance indicators in isolation
- Compiler optimizations are sensitive to the algorithm
- Java/JIT compiled code is significantly faster than JVM interpreted
  - Comparable to optimized C in some cases
- Nothing can fix a dumb algorithm!
Arrays vs. Pointers

- Array indexing involves
  - Multiplying index by element size
  - Adding to array base address

- Pointers correspond directly to memory addresses
  - Can avoid indexing complexity
### Example: Clearing and Array

**clear1**
```c
int clear1(int array[], int size) {
    int i;
    for (i = 0; i < size; i += 1)
        array[i] = 0;
}
```

**clear2**
```c
int clear2(int *array, int size) {
    int *p;
    for (p = &array[0]; p < &array[size];
        p = p + 1)
        *p = 0;
}
```

**Assembly Code 1:**
```assembly
move $t0,$zero   # i = 0
loop1: sll $t1,$t0,2    # $t1 = i * 4
        add $t2,$a0,$t1  # $t2 =
                        #   &array[i]
        sw $zero, 0($t2) # array[i] = 0
        addi $t0,$t0,1   # i = i + 1
        slt $t3,$t0,$a1  # $t3 =
                        #   (i < size)
        bne $t3,$zero,loop1 # if (...) # goto loop1
```

**Assembly Code 2:**
```assembly
move $t0,$a0      # p = & array[0]
    sll $t1,$a1,2  # $t1 = size * 4
    add $t2,$a0,$t1  # $t2 =
        &array[size]
loop2: sw $zero,0($t0) # Memory[p] = 0
        addi $t0,$t0,4   # p = p + 4
        slt $t3,$t0,$t2  # $t3 =
                        #(p<&array[size])
        bne $t3,$zero,loop2 # if (...) # goto loop2
```
Comparison of Array vs. Ptr

- Multiply “strength reduced” to shift
- Array version requires shift to be inside loop
  - Part of index calculation for incremented i
  - c.f. incrementing pointer
- Compiler can achieve same effect as manual use of pointers
  - Induction variable elimination
  - Better to make program clearer and safer
## ARM & MIPS Similarities

- ARM: the most popular embedded core
- Similar basic set of instructions to MIPS

<table>
<thead>
<tr>
<th></th>
<th>ARM</th>
<th>MIPS</th>
</tr>
</thead>
<tbody>
<tr>
<td>Date announced</td>
<td>1985</td>
<td>1985</td>
</tr>
<tr>
<td>Instruction size</td>
<td>32 bits</td>
<td>32 bits</td>
</tr>
<tr>
<td>Address space</td>
<td>32-bit flat</td>
<td>32-bit flat</td>
</tr>
<tr>
<td>Data alignment</td>
<td>Aligned</td>
<td>Aligned</td>
</tr>
<tr>
<td>Data addressing modes</td>
<td>9</td>
<td>3</td>
</tr>
<tr>
<td>Registers</td>
<td>$15 \times 32$-bit</td>
<td>$31 \times 32$-bit</td>
</tr>
<tr>
<td>Input/output</td>
<td>Memory mapped</td>
<td>Memory mapped</td>
</tr>
</tbody>
</table>
Compare and Branch in ARM

- Uses condition codes for result of an arithmetic/logical instruction
  - Negative, zero, carry, overflow
  - Compare instructions to set condition codes without keeping the result
- Each instruction can be conditional
  - Top 4 bits of instruction word: condition value
  - Can avoid branches over single instructions
Instruction Encoding

Chapter 2 — Instructions: Language of the Computer — 86
The Intel x86 ISA

- Evolution with backward compatibility
  - 8080 (1974): 8-bit microprocessor
    - Accumulator, plus 3 index-register pairs
  - 8086 (1978): 16-bit extension to 8080
    - Complex instruction set (CISC)
  - 8087 (1980): floating-point coprocessor
    - Adds FP instructions and register stack
  - 80286 (1982): 24-bit addresses, MMU
    - Segmented memory mapping and protection
  - 80386 (1985): 32-bit extension (now IA-32)
    - Additional addressing modes and operations
    - Paged memory mapping as well as segments
The Intel x86 ISA

- Further evolution...
  - i486 (1989): pipelined, on-chip caches and FPU
    - Compatible competitors: AMD, Cyrix, ...
  - Pentium (1993): superscalar, 64-bit datapath
    - Later versions added MMX (Multi-Media eXtension) instructions
    - The infamous FDIV bug
    - New microarchitecture (see Colwell, The Pentium Chronicles)
  - Pentium III (1999)
    - Added SSE (Streaming SIMD Extensions) and associated registers
  - Pentium 4 (2001)
    - New microarchitecture
    - Added SSE2 instructions
The Intel x86 ISA

And further…

- **AMD64 (2003):** extended architecture to 64 bits
- **EM64T – Extended Memory 64 Technology (2004):**
  - AMD64 adopted by Intel (with refinements)
  - Added SSE3 instructions
- **Intel Core (2006):**
  - Added SSE4 instructions, virtual machine support
- **AMD64 (announced 2007): SSE5 instructions:**
  - Intel declined to follow, instead…
- **Advanced Vector Extension (announced 2008):**
  - Longer SSE registers, more instructions

If Intel didn’t extend with compatibility, its competitors would!

- Technical elegance ≠ market success
# Basic x86 Registers

<table>
<thead>
<tr>
<th>Name</th>
<th>Use</th>
</tr>
</thead>
<tbody>
<tr>
<td>EAX</td>
<td>GPR 0</td>
</tr>
<tr>
<td>ECX</td>
<td>GPR 1</td>
</tr>
<tr>
<td>EDX</td>
<td>GPR 2</td>
</tr>
<tr>
<td>EBX</td>
<td>GPR 3</td>
</tr>
<tr>
<td>ESP</td>
<td>GPR 4</td>
</tr>
<tr>
<td>EBP</td>
<td>GPR 5</td>
</tr>
<tr>
<td>ESI</td>
<td>GPR 6</td>
</tr>
<tr>
<td>EDI</td>
<td>GPR 7</td>
</tr>
<tr>
<td>CS</td>
<td>Code segment pointer</td>
</tr>
<tr>
<td>SS</td>
<td>Stack segment pointer (top of stack)</td>
</tr>
<tr>
<td>DS</td>
<td>Data segment pointer 0</td>
</tr>
<tr>
<td>ES</td>
<td>Data segment pointer 1</td>
</tr>
<tr>
<td>FS</td>
<td>Data segment pointer 2</td>
</tr>
<tr>
<td>GS</td>
<td>Data segment pointer 3</td>
</tr>
<tr>
<td>EIP</td>
<td>Instruction pointer (PC)</td>
</tr>
<tr>
<td>EFLAGS</td>
<td>Condition codes</td>
</tr>
</tbody>
</table>
Basic x86 Addressing Modes

Two operands per instruction

<table>
<thead>
<tr>
<th>Source/dest operand</th>
<th>Second source operand</th>
</tr>
</thead>
<tbody>
<tr>
<td>Register</td>
<td>Register</td>
</tr>
<tr>
<td>Register</td>
<td>Immediate</td>
</tr>
<tr>
<td>Register</td>
<td>Memory</td>
</tr>
<tr>
<td>Memory</td>
<td>Register</td>
</tr>
<tr>
<td>Memory</td>
<td>Immediate</td>
</tr>
</tbody>
</table>

Memory addressing modes

- Address in register
  - Address = \( R_{\text{base}} + \text{displacement} \)
- Address = \( R_{\text{base}} + 2^{\text{scale}} \times R_{\text{index}} \) (scale = 0, 1, 2, or 3)
- Address = \( R_{\text{base}} + 2^{\text{scale}} \times R_{\text{index}} + \text{displacement} \)
### x86 Instruction Encoding

#### Variable length encoding
- Postfix bytes specify addressing mode
- Prefix bytes modify operation
- Operand length, repetition, locking, …

<table>
<thead>
<tr>
<th>Instruction</th>
<th>Format</th>
<th>Description</th>
</tr>
</thead>
<tbody>
<tr>
<td>a. JE EIP + displacement</td>
<td>4 4 8</td>
<td>Condition Displacement</td>
</tr>
<tr>
<td>b. CALL</td>
<td>8</td>
<td>32 Call Offset</td>
</tr>
<tr>
<td>c. MOV EBX, [EDI + 45]</td>
<td>6 1 1 8 8</td>
<td>MOV d w r/m Postbyte Displacement</td>
</tr>
<tr>
<td>d. PUSH ESI</td>
<td>5 3</td>
<td>PUSH Reg</td>
</tr>
<tr>
<td>e. ADD EAX, #6765</td>
<td>4 3 1 32</td>
<td>ADD Reg w Immediate</td>
</tr>
<tr>
<td>f. TEST EDX, #42</td>
<td>7 1 8 32</td>
<td>TEST w Postbyte Immediate</td>
</tr>
</tbody>
</table>
Implementing IA-32

- Complex instruction set makes implementation difficult
  - Hardware translates instructions to simpler microoperations
    - Simple instructions: 1–1
    - Complex instructions: 1–many
  - Microengine similar to RISC
  - Market share makes this economically viable
- Comparable performance to RISC
  - Compilers avoid complex instructions
ARM v8 Instructions

- In moving to 64-bit, ARM did a complete overhaul
- ARM v8 resembles MIPS
  - Changes from v7:
    - No conditional execution field
    - Immediate field is 12-bit constant
    - Dropped load/store multiple
    - PC is no longer a GPR
    - GPR set expanded to 32
    - Addressing modes work for all word sizes
    - Divide instruction
    - Branch if equal/branch if not equal instructions
Fallacies

- Powerful instruction $\Rightarrow$ higher performance
  - Fewer instructions required
  - But complex instructions are hard to implement
    - May slow down all instructions, including simple ones
  - Compilers are good at making fast code from simple instructions

- Use assembly code for high performance
  - But modern compilers are better at dealing with modern processors
  - More lines of code $\Rightarrow$ more errors and less productivity
Fallacies

- Backward compatibility $\Rightarrow$ instruction set doesn’t change
  - But they do accrete more instructions

![Graph showing the growth of x86 instruction set over years]

x86 instruction set
Pitfalls

- Sequential words are not at sequential addresses
  - Increment by 4, not by 1!
- Keeping a pointer to an automatic variable after procedure returns
  - e.g., passing pointer back via an argument
  - Pointer becomes invalid when stack popped
Concluding Remarks

- Design principles
  1. Simplicity favors regularity
  2. Smaller is faster
  3. Make the common case fast
  4. Good design demands good compromises

- Layers of software/hardware
  - Compiler, assembler, hardware

- MIPS: typical of RISC ISAs
  - c.f. x86
Concluding Remarks

- Measure MIPS instruction executions in benchmark programs
  - Consider making the common case fast
  - Consider compromises

<table>
<thead>
<tr>
<th>Instruction class</th>
<th>MIPS examples</th>
<th>SPEC2006 Int</th>
<th>SPEC2006 FP</th>
</tr>
</thead>
<tbody>
<tr>
<td>Arithmetic</td>
<td>add, sub, addi</td>
<td>16%</td>
<td>48%</td>
</tr>
<tr>
<td>Data transfer</td>
<td>lw, sw, lb, lbu, lh, lhu, sb, lui</td>
<td>35%</td>
<td>36%</td>
</tr>
<tr>
<td>Logical</td>
<td>and, or, nor, andi, ori, sll, srl</td>
<td>12%</td>
<td>4%</td>
</tr>
<tr>
<td>Cond. Branch</td>
<td>beq, bne,slt, slti, sltiu</td>
<td>34%</td>
<td>8%</td>
</tr>
<tr>
<td>Jump</td>
<td>j, jr, jal</td>
<td>2%</td>
<td>0%</td>
</tr>
</tbody>
</table>